# Maintainer:

pkgname=perl-devel-patchperl
pkgver=2.08
pkgrel=2
pkgdesc="Patch perl source a la Devel::PPPort's buildperl.pl"
arch=('any')
url="https://search.cpan.org/dist/Devel-PatchPerl/"
license=('PerlArtistic' 'GPL')
depends=('perl' 'perl-file-pushd' 'perl-module-pluggable')
options=('!emptydirs')
source=("https://search.cpan.org/CPAN/authors/id/B/BI/BINGOS/Devel-PatchPerl-$pkgver.tar.gz")
sha256sums=('69c6e97016260f408e9d7e448f942b36a6d49df5af07340f1d65d7e230167419')

build() {
  cd Devel-PatchPerl-$pkgver
  perl Makefile.PL INSTALLDIRS=vendor
  make
}

check() {
  cd Devel-PatchPerl-$pkgver
  make test
}

package() {
  cd Devel-PatchPerl-$pkgver
  make install DESTDIR="$pkgdir"
}
